При работе в СЭД Directum часто возникает необходимость формирования отчетов по какому-либо определенному виду электронных документов, созданных за определенный период. Для отчетов основными требованиями можно выделить:
1. Информативность – в отчете должны быть представлены данные, специфичные для конкретного, выбранного, типа электронных документов.

2. Скорость работы – формирование отчета не должно занимать длительного промежутка времени.

3. Презентабельный вид – результирующий отчет должен быть отформатированным, все ячейки должны соответствовать типам данных.

4. Возможность записи параметров отчета

Решение

Что бы сотрудникам не приходилось каждый раз заново формировать параметры отчета было принято решение создать специальный справочник, в котором бы хранились эти данные. 
Реквизиты справочника:
	Раздел
	Имя
	Заголовок
	Тип

	Карточка
	Дата
	Период с
	Дата

	Карточка
	Дата2
	Период по
	Дата

	Карточка
	SNGBLongString
	Список реквизитов
	Строка 4096

	Карточка
	SNGBLongString2
	Отображаемые реквизиты
	Строка 4096

	Карточка
	String
	Наименование ТЭД
	Строка

	Карточка
	ShortString2
	Код ТЭД
	Строка

	Таблица
	СтрокаТ
	Код ТЭД
	Строка

	Таблица
	ДопТ
	Наименование реквизита
	Строка


Выборка для наименование ТЭД имеет вид:

// Выбираем виды электронных документов, которые попадают в отчет.
AddWhere = "(MBAnalit.Prim like '%POS_ЗНВТ%' or 

             MBAnalit.Prim like '%POS_ЗНУТ%' or 

             MBAnalit.Prim like '%POS_ЗНСТ%' or 

             MBAnalit.Prim like '%POS_EC_ЗНВТ%' or

             MBAnalit.Prim like '%POS_EC_ЗНУТ%')"   

if SelectMode = 1 

  AddLike = InputValue

else

  AddLike = ""

endif  

TempRef = Справочник('ТЭД';;;;AddWhere;"Выбор";;;;;;;;;AddLike;)

if TempRef <<>> 'НЕУДАЧА: ESC!'

  ReferenceRecord = References.ТЭД.GetObjectByCode(TempRef)

  Requisite.Value = ReferenceRecord.Наименование
  Object.ShortString2 = ReferenceRecord.Код
//Получаем реквизиты электронного документа
  !TypeEDoc = ReferenceRecord.Код
  !Component = GetComponent()

  !CurrentLanguage = !Component.Application.Localization.CurrentLanguage

  !Запрос = "select

               d.!SYSREF_EDOCUMENT_CARD_REQUISITES.РеквКод!

             from 

               !SYSREF_EDOCUMENT_CARD_REQUISITES! d

             where 

               d.!SYSREF_EDOCUMENT_CARD_REQUISITES.Вид! = !ШАД.ComponentID!

               and d.!SYSREF_EDOCUMENT_CARD_REQUISITES.РеквКод! <> 'ВидДокумента'

               and d.!SYSREF_EDOCUMENT_CARD_REQUISITES.Раздел! = 'Ш'  -- находиться в разделе карточка

               and d.Regim <> 'С' --не скрытый"

  !Запрос = PreprocQuery(!Запрос) 

  !ListReqSh = SQL(!Запрос; ; "','")

  !ListReqSh = "'" & !ListReqSh & "'"

  !ListAddReq = ""

  !Запрос = "select

                 v.!SYSREF_EDOCUMENT_CARD_TYPES.Наименование!, r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqFieldName!, 

                 isnull(Replace(loc.!SYSREF_LOCALIZATION_DICTIONARY.ISBViewString!, char(13) + char(10), ''), 

                 d.!SYSREF_EDOCUMENT_CARD_REQUISITES.Наименование!) as Name, 

                 r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqType!,

                 r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqLength!, r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqPrecision!,

                 r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqTokens!, VidAn.!SYSREF_REFERENCES.Код!, 

                 r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqFormat!, 

                 r.!SYSREF_EDOCUMENT_REQUISITES.Код!

               from

                 !SYSREF_EDOCUMENT_CARD_REQUISITES! d,

                 !SYSREF_EDOCUMENT_REQUISITES! r,

                 !SYSREF_EDOCUMENT_CARD_TYPES! v,

                 !SYSREF_LOCALIZATION_DICTIONARY! loc,

                 !SYSREF_REFERENCES! VidAn

               where 

                 ltrim(rtrim(v.!SYSREF_EDOCUMENT_CARD_TYPES.Код!)) = '" & Trim(!TypeEDoc) & "'

                 and d.!SYSREF_EDOCUMENT_CARD_REQUISITES.Вид! = v.!SYSREF_EDOCUMENT_CARD_TYPES.ИД!

                 and d.!SYSREF_EDOCUMENT_CARD_REQUISITES.Раздел! = 'Ш'  -- карточка
                 and r.!SYSREF_EDOCUMENT_REQUISITES.Код! = d.!SYSREF_EDOCUMENT_CARD_REQUISITES.РеквКод!

                 and d.!SYSREF_EDOCUMENT_CARD_REQUISITES.РеквКод! not in (" & !ListReqSh & ")  -- НЕ совпадающие с рекв.шаблона
                 and r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqType! <> 'М'  -- НЕ текст
                 and r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqGenerated! = 'Е'

                 and r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqReference! *= 

                   VidAn.!SYSREF_REFERENCES.ComponentIDField!

                 and left(d.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!, 

                   charindex('.', d.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!) - 1) *= 

                   loc.!SYSREF_LOCALIZATION_DICTIONARY.ISBViewGroupCode!

                 and right(d.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!, 

                   Len(d.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!) -

                   charindex('.', d.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!)) *= 

                   loc.!SYSREF_LOCALIZATION_DICTIONARY.Код!

                 and loc.!SYSREF_LOCALIZATION_DICTIONARY.ISBViewLangCode! = '" & !CurrentLanguage & "' 

               order by

                 d.!SYSREF_EDOCUMENT_CARD_REQUISITES.РеквНомер!" 

    !Запрос = PreprocQuery(!Запрос)             

    ForEach !CurRecord=CSQL(!Запрос;;"!")         // Меняем на "!" т.к. в Признаке встречается ";"

      If !CurRecord<<>>""  

        !Requisite=SubString(!CurRecord;'!';3)

        !ListAddReq=AddSubString(!Requisite;!ListAddReq;"|")

      EndIf

    EndForEach

    Object.SNGBLongString = !ListAddReq

    DDS = Object.DetailDataSet(1)

    DDS.First

    Out = ""

    foreach Element in DDS

      SelectedDocGroupeName = Element.Requisites("ДопТ").DisplayText

      ElementType = Element.СтрокаТ
      if Object.ShortString2 == ElementType

        if Out <<>> "" 

          Out = Out & "|" & SelectedDocGroupeName

        else

          Out = SelectedDocGroupeName 

        endif 

      endif 

    endforeach

    if Out <<>> ""

      Object.SNGBLongString2 = Out

    else

      Object.SNGBLongString2 = Object.SNGBLongString

    endif    

endif
Для выбора реквизитов создадим специальную кнопку «…»:

List = Object.SNGBLongString

if Not Assigned(List)

  Text = "Необходимо выбрать ТЭД"

  exit(Text)

Endif
StrList = CreateStringList()

StrList.Delimiter = '|'

StrList.DelimitedText = List

StrList.Sorted = true

List = StrList.DelimitedText

DualList = CreateDualListDialog()

DualList.Title = "Параметры отчета"

DualList.AvailableItemsTitle = 'Доступные параметры'

DualList.SelectedItemsTitle = 'Выбранные параметры'

DDS = Object.DetailDataSet(1)

DDS.First

isEmpty = true

foreach Element in DDS

  ElementType = Element.СтрокаТ

  if Object.ShortString2 == ElementType

    isEmpty = false

    exitfor   

  endif

endforeach

if not isEmpty

  foreach str in CSubString(List;"|")

    DualList.AvailableItems.Add(str)       

  endforeach

  foreach Element in DDS

     SelectedDocGroupeName = Element.Requisites("ДопТ").DisplayText

     ElementType = Element.СтрокаТ

     if Object.ShortString2 == ElementType   

       DualList.SelectedItems.Add(SelectedDocGroupeName)

       Index = DualList.AvailableItems.IndexOf(SelectedDocGroupeName)

       if Index > -1

          DualList.AvailableItems.Delete(Index)

       endif

     endif  

  endforeach 

else

  foreach str in CSubString(List;"|")

    DualList.SelectedItems.Add(str)

  endforeach

endif

DualList.Execute() 

// Все выбранные значения записать в строку-информацию 

DualList.SelectedItems.Delimiter = "|"

Data = DualList.SelectedItems.DelimitedText 

if Assigned(Data)

  Object.SNGBLongString2 = Data 

 Detail = Object.DetailDataSet(1)

 Detail.DisableControls

 Detail.Events.DisableAll

 Count = Detail.RecordCount

 Detail.First

 while not Detail.EOF

   if Object.ShortString2 == Detail.СтрокаТ

      Detail.Delete

   else

      Detail.Next

   endif

 endwhile

 Detail.Events.EnableAll

 Detail.EnableControls

  DDS = Object.DetailDataSet(1)

  foreach str in CSubString(Data;"|")

    DDS.Append

    DDS.ДопТ = str

    DDS.СтрокаТ = Object.ShortString2

  endforeach
endif 

DDS.First       

И реализуем кнопку поиска, вызывающую отчет с необходимыми параметрами:

if Not Assigned(Object.SNGBLongString)

  Text = "Необходимо выбрать ТЭД"

  exit(Text)

Endif
CurRep = Reports.GetObjectByName("SNGB_POS_Отчет_по_заявкам_эквайринга")

RepParam = CurRep.Params

RepParam.Add("DateBegin";Object.Дата)

RepParam.Add("DateEnd";Object.Дата2)

RepParam.Add("TypeDoc";Object.ShortString2)

RepParam.Add("Filter";Object.SNGBLongString2)

CurRep.Execute
Отчет. Приложение для просмотра отчета имеет расширение XML, тип редактора - встроенный. 
Расчет:
  RepParam = Object.Params
  if RepParam.Count = 0

    //БЛОК СОЗДАНИЕ ЗАПРОСА

    // Определим наименование записи - это имя текущего пользователя + параметр !Результат,

    НаимАн = ПользовательИмя()

    Reference = References.ReferenceFactory('SNGB_POS_Search').GetComponent

    TypeReferenceID = Reference.ComponentID 

    // Определим, есть ли для данного пользователя запись спроавочника.

    КодАн = ПоискСпр('SNGB_POS_Search'; "Наименование:" & НаимАн)

    // Если для пользователя нет записи справочника, то создадим его.

    Если КодАн == ""

      КодАн = СпрСоздать('SNGB_POS_Search'; "АВТОНОМЕР"; "ВСЕ";;)

      СпрРекв('SNGB_POS_Search'; КодАн; "Наименование";; НаимАн)

      КодАн = СпрСохранить('SNGB_POS_Search'; КодАн; 'ВСЕ';; "Код")

    КонецЕсли

    // Откроем карточку записи.

    FindDataSet = GetReferenceRecord('SNGB_POS_Search'; КодАн)

    FindComp = FindDataSet.Component

    FindComp.Form.ShowModal 

    Exit()

  endif

  Prog = CreateProgress()

  Prog.Text = 'Построение параметров запроса'

  Prog.Show  

  !Component = GetComponent()

  !CurrentLanguage = !Component.Application.Localization.CurrentLanguage

  // По справочнику Типы электронных документов

  !IDTypeRefTKD=ИДТипСпр("ТЭД")

  !IDTypeRefSHB=ИДТипСпр("ШАД")  //Шаблоны электронных документов - отсюда берем основные рек-ты

  !ТАБ=TAB
  // Загрузить строки локализации

  !EDocTypeStr = LoadString('ISB_COLUMN_Г000012'; 'ISBSYSDEV') // Тип карточки эл. документа

  !AuthStr = LoadString('REQ_Д000007'; 'EDM') // Автор

  !AppStr = LoadString('REQ_Д000009'; 'EDM') // Приложение                      

  !CreatedFromStr = LoadString('DIRLABEL_6209AF7F-9EAA-4343-8C24-5CEAC98C580C'; 'COMMON') // Дата создания c

  !CreatedToStr = LoadString('DIRLABEL_35C7217E-378C-4BAC-8FF6-03F499D8C1AE'; 'COMMON') // Дата создания по

  !ModFromStr = LoadString('DIRLABEL_3F43A157-06E3-4E71-B749-4136CAA1CE99'; 'COMMON') // Дата редактирования с

  !ModToStr = LoadString('DIRLABEL_C26ED0F4-6DEC-4C37-9378-520884038E75'; 'COMMON') // Дата редактирования по

  !Author = "" // Добавить фильтр по автору

  !DocEditor = ""// Добавить фильтр по изменениям

  !CreateDate1 = RepParam.ValueByName("DateBegin")

  !CreateDate2 = RepParam.ValueByName("DateEnd")

  !TypeEDoc = RepParam.ValueByName("TypeDoc")

  !Filter = RepParam.ValueByName("Filter") 

  //=============================================================================

  //-------- Найдем реквизиты шаблона

  !Запрос = "select

               d.!SYSREF_EDOCUMENT_CARD_REQUISITES.РеквКод!

             from 

               !SYSREF_EDOCUMENT_CARD_REQUISITES! d

             where 

               d.!SYSREF_EDOCUMENT_CARD_REQUISITES.Вид! = !ШАД.ComponentID!

               and d.!SYSREF_EDOCUMENT_CARD_REQUISITES.РеквКод! <> 'ВидДокумента'

               and d.!SYSREF_EDOCUMENT_CARD_REQUISITES.Раздел! = 'Ш'  -- находиться в разделе карточка

               and d.Regim <> 'С' --не скрытый"

  !Запрос = PreprocQuery(!Запрос) 

  !ListReqSh = SQL(!Запрос; ; "','")

  !ListReqSh = "'" & !ListReqSh & "'"     

  //-------- Найдем доп-е реквизиты выбранного типа док. и запросим их --------

  // Попутно сформируем select для формирования запроса по созданию и заполнению таблицы 

  //   для отчета, а также для рекв-аналитик where и from с MBAnalit для окончательного запроса

  !Type = ""

  !NameTypeEDoc = !Type

  !ListTypeRef = !Type

  !ListTypeAddReq = !Type

  !ListAddReq = !Type

  !ListCodeRequisite=""

  !ListNameField=""

  !ListTypeReq=""

  !DopNameColumn=""

  !ListPickReq=""

  !ListPriznValue=""

  !ListShortValuePickReq=""

  !AddFldCrTable=""  // Для запроса по созданию таблицы

  !AddSelect=""     // Для запроса по заполнению таблицы 

  !Where = ""  // Для последнего запроса (!From и !Where- для рек. типа Аналитика)

  !From = !Where
  !Select = !Where
  !NumReq=0       // сколько полей нужно добавить в таблицу для отчета

  If !TypeEDoc<<>>""

//    !TypeEDoc=ФмтСтр("10|";!TypeEDoc)

    !Запрос = "select

                 v.!SYSREF_EDOCUMENT_CARD_TYPES.Наименование!, r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqFieldName!, 

                 isnull(Replace(loc.!SYSREF_LOCALIZATION_DICTIONARY.ISBViewString!, char(13) + char(10), ''), 

                 d.!SYSREF_EDOCUMENT_CARD_REQUISITES.Наименование!) as Name, 

                 r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqType!,

                 r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqLength!, r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqPrecision!,

                 r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqTokens!, VidAn.!SYSREF_REFERENCES.Код!, 

                 r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqFormat!, 

                 r.!SYSREF_EDOCUMENT_REQUISITES.Код!

               from

                 !SYSREF_EDOCUMENT_CARD_REQUISITES! d,

                 !SYSREF_EDOCUMENT_REQUISITES! r,

                 !SYSREF_EDOCUMENT_CARD_TYPES! v,

                 !SYSREF_LOCALIZATION_DICTIONARY! loc,

                 !SYSREF_REFERENCES! VidAn

               where 

                 ltrim(rtrim(v.!SYSREF_EDOCUMENT_CARD_TYPES.Код!)) = '" & Trim(!TypeEDoc) & "'

                 and d.!SYSREF_EDOCUMENT_CARD_REQUISITES.Вид! = v.!SYSREF_EDOCUMENT_CARD_TYPES.ИД!

                 and d.!SYSREF_EDOCUMENT_CARD_REQUISITES.Раздел! = 'Ш'  -- карточка

                 and r.!SYSREF_EDOCUMENT_REQUISITES.Код! = d.!SYSREF_EDOCUMENT_CARD_REQUISITES.РеквКод!

                 and d.!SYSREF_EDOCUMENT_CARD_REQUISITES.РеквКод! not in (" & !ListReqSh & ")  -- НЕ совпадающие с рекв.шаблона

                 and r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqType! <> 'М'  -- НЕ текст

                 and r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqGenerated! = 'Е'

                 and r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqReference! *= 

                   VidAn.!SYSREF_REFERENCES.ComponentIDField!

                 and left(d.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!, 

                   charindex('.', d.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!) - 1) *= 

                   loc.!SYSREF_LOCALIZATION_DICTIONARY.ISBViewGroupCode!

                 and right(d.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!, 

                   Len(d.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!) -

                   charindex('.', d.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!)) *= 

                   loc.!SYSREF_LOCALIZATION_DICTIONARY.Код!

                 and loc.!SYSREF_LOCALIZATION_DICTIONARY.ISBViewLangCode! = '" & !CurrentLanguage & "' 

               order by

                 d.!SYSREF_EDOCUMENT_CARD_REQUISITES.РеквНомер!" 

    !Запрос = PreprocQuery(!Запрос)     

    TableColumnType = "Title"

    TableColumnLength = 120   

    SymbolLength = 6 

    Records = SQL(!Запрос;;;"!")   

    foreach str in CSubString(!Filter;"|")

      i = 0     

      !Requisite = ""

      while (str <<>> !Requisite) and (i <= SubStringCount(Records;"|") )

        i = i + 1

        !CurRecord = SubString(Records;"|";i)

        If !CurRecord<<>>""  

          !Requisite=SubString(!CurRecord;'!';3)

        else

          !Requisite=""

        endif        

      endwhile  

      if  (str == !Requisite)  

        !NameTypeEDoc=SubString(!CurRecord;'!';1)

        !Field=SubString(!CurRecord;'!';2) 

        !TypeRequisite=Trim(SubString(!CurRecord;'!';4))

        !CodeRequisite=Trim(SubString(!CurRecord;'!';10)) 

        !ListNameField=AddSubString(!Field;!ListNameField;"|")

        !ListCodeRequisite=AddSubString(!CodeRequisite;!ListCodeRequisite;"|")

        !ListTypeReq=AddSubString(!TypeRequisite;!ListTypeReq;"|")  

        !NumReq=!NumReq+1     //номер доп.реквизита, а в конце- кол-во доп-х рекв.

        !DopNameColumn=!DopNameColumn&!ТАБ&!Requisite //названия колонок для доп-х реквизитов

         // Разные типы реквизитов

        If !TypeRequisite=="А"  // аналитика

          !CodeTypeRef=SubString(!CurRecord;'!';8)

          !Type="Аналитика:"&!CodeTypeRef

          !AddFldCrTable=!AddFldCrTable&", Recv"&!NumReq&" int null "  // Тип поля для создания таблицы для отчета

          !AddSelect=!AddSelect&", a."&!Field  // выбор поля при заполнении таблицы

          !From=!From&" , MBAnalitSpr a"&!NumReq     // добавим MBAnalit для последнего запроса

          !Where=!Where&" and t.Recv"&!NumReq&"*=a"&!NumReq&".Analit "

          !Select=!Select&" ,a"&!NumReq&".NameAn "   // для последнего запроса

          TableColumnType = AddSubString("String"; TableColumnType;  "|")

          ColumnLength = Length(!Requisite)*SymbolLength

          TableColumnLength = AddSubString(ColumnLength; TableColumnLength; "|")

        Else
          !Select=!Select&", t.Recv"&!NumReq   // для последнего запроса

          If !TypeRequisite=="П"  // признак,где-то надо взять короткий,где-то полный(но без &)

            !ListPickReq=AddSubString(!Requisite;!ListPickReq;"|")

            !PriznValue=SubString(!CurRecord;'!';7)

            !ListPriznValue = ''

            !МахLen=0  // для длины поля при создании таблицы

            !PriznShortValue=""   // символ для выбора длинного значения

            LongValueList = ""  

            !PriznValue=Replace(!PriznValue;";";",")

            foreach Value = CSubString(!PriznValue; ",")

              LocValue = SubString(Value; "|"; 2)

              if Copy(LocValue; 1; 1) == "@"

                LocValue = Copy(LocValue; 2; Length(LocValue) - 1)

                LongValue = LoadString(SubString(LocValue; "."; 2); SubString(LocValue; "."; 1))

              else

                LongValue = LocValue

              endif    

              LongValueList = AddSubString(LongValue; LongValueList; "|")

              ShortValue = SubString(Value; "="; 1)

              !PriznShortValue=AddSubString(ShortValue; !PriznShortValue; ";")

              !ListPriznValue = AddSubString(LongValue; !ListPriznValue; ',')

              !МахLen=Макс(!МахLen;Length(Value))

            endforeach 

            !ListShortValuePickReq=AddSubString(!PriznShortValue;!ListShortValuePickReq;"|")

            !Type="Признак:"& !ListPriznValue

            !AddFldCrTable=!AddFldCrTable&", Recv"&!NumReq&" varchar("&!МахLen&") null "  // Тип поля для создания таблицы для отчета

           // сформируем заполнение поля полным признаком по его короткому значению

            !CurEl=1

            !SelPrizn=""

            ForEach !LongValue=CSubString(!PriznValue;",")  //без знака &

              !ShortValue=SubString(!PriznShortValue;";";!CurEl) 

              LongValue = SubString(LongValueList; "|"; !CurEl)

              !SelPrizn=!SelPrizn&" when a."&!Field&"='"&!ShortValue&"' then '"&LongValue&"'"

              !CurEl=!CurEl+1

            EndForEach  

            !AddSelect=!AddSelect&" , Case "&!SelPrizn&" end"  // выбор поля при заполнении таблицы   

            TableColumnType = AddSubString("String"; TableColumnType;  "|")

            ColumnLength = Length(!Requisite)*SymbolLength

            TableColumnLength = AddSubString(ColumnLength ; TableColumnLength; "|") 

          Else  // Остальные типы рекв. ~ одинаково

            //!СтрУ=AddSubString(!Рекв;!СтрУ;"|")

            If !TypeRequisite=="С"

              !Len=SubString(!CurRecord;'!';5)

              !Type="Строка:"&!Len
              If SubString(!CurRecord;'!';9)=="П"  // уберем слева пробелы, если выравнивание вправо

                !AddSelect=!AddSelect&",Ltrim(a."&!Field&")"  // выбор поля при заполнении таблицы

              Else
                !AddSelect=!AddSelect&", a."&!Field  // выбор поля при заполнении таблицы

              EndIf
              !AddFldCrTable=!AddFldCrTable&", Recv"&!NumReq&" varchar("&!Len&") null "  // Тип поля для создания таблицы для отчета

              TableColumnType = AddSubString("String";TableColumnType;  "|")

              ColumnLength = Length(!Requisite)*SymbolLength

              TableColumnLength = AddSubString(ColumnLength ; TableColumnLength; "|")   

            EndIf

            If !TypeRequisite=="Д"

              !Type="Дата"

              !AddFldCrTable=!AddFldCrTable&", Recv"&!NumReq&" datetime null "  // Тип поля для создания таблицы для отчета

              !AddSelect=!AddSelect&", a."&!Field  // выбор поля при заполнении таблицы

              TableColumnType = AddSubString( "DateTime"; TableColumnType;"|")

              ColumnLength = Length(!Requisite)*SymbolLength

              TableColumnLength = AddSubString(ColumnLength ; TableColumnLength; "|")   

            EndIf

            If !TypeRequisite=="Ч"

              !Toch=SubString(!CurRecord;'!';6)

              !Type="Число:"&!Toch

              !AddFldCrTable=!AddFldCrTable&", Recv"&!NumReq&" float null "  // Тип поля для создания таблицы для отчета

              !AddSelect=!AddSelect&", a."&!Field  // выбор поля при заполнении таблицы

              TableColumnType = AddSubString( "Number"; TableColumnType;"|")

              ColumnLength = Length(!Requisite)*SymbolLength

              TableColumnLength = AddSubString(ColumnLength;TableColumnLength;  "|")  

            EndIf

            If !TypeRequisite=="Ц"

              !Type="Число:0"

              !AddFldCrTable=!AddFldCrTable&", Recv"&!NumReq&" int null "  // Тип поля для создания таблицы для отчета

              !AddSelect=!AddSelect&", a."&!Field  // выбор поля при заполнении таблицы              

              TableColumnType = AddSubString( "Number";TableColumnType; "|")

              ColumnLength = Length(!Requisite)*SymbolLength

              TableColumnLength = AddSubString(ColumnLength; TableColumnLength; "|")               

            EndIf

          EndIf   // П 

        EndIf     // А

        !ListAddReq=AddSubString(!Requisite;!ListAddReq;"|")

        !ListTypeAddReq=AddSubString(!Type;!ListTypeAddReq;"|")

      endif              

    endforeach   

    If !ListAddReq<<>>""  //выбрали тип док. и в нем есть доп. рекв.

      !ListAddReqValue= ""//InputDialog(!ListAddReq;;!ListTypeAddReq;!NameTypeEDoc)

    EndIf              

  Else   

    !NumReq=1

    !Запрос = "select

                 distinct ra.!SYSREF_EDOCUMENT_CARD_REQUISITES.РеквКод!, 

                 r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqFieldName!, 

                 an.!SYSREF_REFERENCES.ComponentIDField!, 

                 isnull(loc.!SYSREF_LOCALIZATION_DICTIONARY.ISBViewString!, 

                   ra.!SYSREF_EDOCUMENT_CARD_REQUISITES.Наименование!) as Name

               from

                 !SYSREF_EDOCUMENT_CARD_REQUISITES! ra,

                 !SYSREF_EDOCUMENT_REQUISITES! r,

                 !SYSREF_REFERENCES! an,

                 !SYSREF_LOCALIZATION_DICTIONARY! loc

               where 

                 ra.!SYSREF_EDOCUMENT_CARD_REQUISITES.Раздел! = 'Ш'

                 and r.!SYSREF_EDOCUMENT_REQUISITES.Код! = ra.!SYSREF_EDOCUMENT_CARD_REQUISITES.РеквКод!

                 and r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqType! = 'А' -- только справочники

                 and r.!SYSREF_EDOCUMENT_REQUISITES.Код! not in ('Пользователь2', 'Пользователь', 'ТекстРедактор2')

                 and r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqGenerated! = 'Е'

                 and r.!SYSREF_EDOCUMENT_REQUISITES.ISBRefReqReference! *= an.!SYSREF_REFERENCES.ComponentIDField!

                 and left(ra.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!, 

                   charindex('.',ra.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!)-1) *= 

                   loc.!SYSREF_LOCALIZATION_DICTIONARY.ISBViewGroupCode!

                 and right(ra.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!, 

                   Len(ra.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!) - 

                   charindex('.',ra.!SYSREF_EDOCUMENT_CARD_TYPES.Table.ISBEDocTypeReqDescLocalizeID!)) *= 

                   loc.!SYSREF_LOCALIZATION_DICTIONARY.Код!

                 and loc.!SYSREF_LOCALIZATION_DICTIONARY.ISBViewLangCode! = '" & !CurrentLanguage & "' "    

    !Запрос = PreprocQuery(!Запрос)

    ForEach !CurRecord=CSQL(!Запрос;;"!")         // Меняем на "!" т.к. в Признаке встречается ";"   

      !CodeRequisite=SubString(!CurRecord;'!';1)

      !Field=SubString(!CurRecord;'!';2)

      !CodeTypeRef=SubString(!CurRecord;'!';3)

      !NameRequisite=Trim(SubString(!CurRecord;'!';4))

      !DopNameColumn=!DopNameColumn&!ТАБ&!NameRequisite //названия колонок для доп-х реквизитов

     // Разные типы реквизитов

      !From=!From&" , MBAnalitSpr a"&!NumReq     // добавим MBAnalit для последнего запроса

      !Where=!Where&" and t.Recv"&!NumReq&"*=a"&!NumReq&".Analit "

      !Select=!Select&" ,a"&!NumReq&".NameAn "   // для последнего запроса

      !AddFldCrTable=!AddFldCrTable&", Recv"&!NumReq&" int null "  // Тип поля для создания таблицы для отчета

      !AddSelect=!AddSelect&", a."&!Field  // выбор поля при заполнении таблицы

      !NumReq=!NumReq+1       

    EndForEach 

  EndIf
  //======================= Сформируем ограничения =============================

  //------------ По основным реквизитам

  !AddWhere=""   

  If !TypeEDoc<<>>""

    !IDTypeEDoc=ИДТипСпр(!TypeEDoc)  

    !AddWhere=AddSubString("a.!ПЭА.ComponentIDField!="&!IDTypeEDoc;!AddWhere;" and ")

  EndIf

  If !Author<<>>""

    !IDAuthor=СпрПолеЗнач("ПОЛ";!Author;"Analit")

    !AddWhere=AddSubString("a.!ПЭА.ISBEDocAuthor!="&!IDAuthor;!AddWhere;" and ")

  EndIf

  If !DocEditor<<>>""

    !IDDocEditor=СпрПолеЗнач("ПРР";!DocEditor;"Analit")

    !AddWhere=AddSubString("a.!ПЭА.ISBEDocEditor!="&!IDDocEditor;!AddWhere;" and ")

  EndIf

  If Assigned(!CreateDate1)

    !AddWhere=AddSubString("a.!ПЭА.ISBEDocCreateDate!>='"&!CreateDate1&"'";!AddWhere;" and ")

  EndIf

  If Assigned(!CreateDate2)

    !AddWhere=AddSubString("a.!ПЭА.ISBEDocCreateDate!<='"&!CreateDate2&" 23:59:59'";!AddWhere;" and ")

  EndIf
  !AddFrom=" "

  //------------ По дополнительным реквизитам

  !СтрокаДопУсл=""

  !Номер=0

  If !ListAddReq<<>>""  // заданы фильтры по доп.рекв.

    !n=1

    ForEach !Value=CSubString(!ListAddReqValue;"|")

      If Trim(!Value)<<>>""  // задали в фильтре

        !Field=SubString(!ListNameField;"|";!n)  // прочитаем условие

        !Type=SubString(!ListTypeAddReq;"|";!n)

        !TypeReq=SubString(!Type;":";1)

       // В зависимости от типа реквизита - разные методы

        If !TypeReq=="Дата" or !TypeReq=="Строка"

          !AddWhere=AddSubString("a."&!Field&"='"&!Value&"'";!AddWhere;" and ")

        EndIf

        If !TypeReq=="Число"

          !AddWhere=AddSubString("a."&!Field&"="&!Value;!AddWhere;" and ")

        endif

        If !TypeReq=="Аналитика"

          !TypeRefReq=SubString(!Type;":";2)

          !IDValue=СпрПолеЗнач(!TypeRefReq;!Value;"Analit")

          !AddWhere=AddSubString("a."&!Field&"="&!IDValue;!AddWhere;" and ")

        endif  

        If !TypeReq=="Признак"

          !ListValue=SubString(!Type;":";2)

          !NumVal=FindSubString(!Value;!ListValue;",")

          !NumPickReq=FindSubString(SubString(!ListAddReq;"|";!n);!ListPickReq;"|")

          !ShortValue=SubString(!ListShortValuePickReq;"|";!NumPickReq)

          !Value=SubString(!ShortValue;";";!NumVal)

          !AddWhere=AddSubString("a."&!Field&"='"&!Value&"'";!AddWhere;" and ")

        EndIf

      EndIf

      !n=!n+1

    КонецВсе

  EndIf

  if !AddWhere<<>>""

    !AddWhere=" and "&!AddWhere

  EndIf  

  Prog.Text = 'Формирование временной таблицы'

  // получить имя пользователя БД

  !UserName = ПользовательИмя('SQL')

  //=============================================================================        

  !TblName2 = "["&!UserName&"].tmpSpis"

  // Создадим таблицу для отчета

  УдалитьТаблицу("tmpSpis")

  SQL("create table " & !TblName2 & "(Analit int null,

                            TipD varchar(50) null,  --Тип ЭД

                            Dok varchar(270) null,  --Наименование ЭД

                            Avtor int null,         --Автор

                            Pril int null,          --Приложение

                            DataN datetime null,    --Дата создания

                            DataRed datetime null,  --Дата последнего редактирования

                            DataV datetime null,

                            KolVer int null,

                            Size int null"

                            & !AddFldCrTable & ")")  //дополнительные реквизиты

  // Заполним таблицу для отчета

  !Запрос = "Insert into " & !TblName2 & "

             select 

               a.!ПЭА.ИД!, t.!SYSREF_EDOCUMENT_CARD_TYPES.Код!, 

               a.!ПЭА.ISBEDocName! + ' (' + convert(varchar,a.!ПЭА.ИД!) + ')', a.!ПЭА.ISBEDocAuthor!, 

               a.!ПЭА.ISBEDocEditor!, a.!ПЭА.ISBEDocCreateDate!, a.!ПЭА.ISBEDocModifyDate!, a.!ПЭА.ISBEDocExtractDate!, 

               (select 

                  count(ver.!ПЭА.Versions.ИД!)

                from 

                  !ПЭА.Versions! ver

                where 

                  ver.!EDocLinkFieldName! = a.!ПЭА.ИД!),

               (select 

                  sum(ver.!ПЭА.Versions.ISBVersionSize!)

                from 

                  !ПЭА.Versions! ver

                where 

                  ver.!EDocLinkFieldName!=a.!ПЭА.ИД!) " & !AddSelect & "

             from !ПЭА! a, !SYSREF_EDOCUMENT_CARD_TYPES! t" & !AddFrom & " 

             where  

               t.!SYSREF_EDOCUMENT_CARD_TYPES.ИД! = a.!ПЭА.ComponentIDField!" & !AddWhere

  !Запрос = PreprocQuery(!Запрос)          

  SQL(!Запрос)

  If SQL("select 1 from "&!TblName2)==""

    !Msg = LoadString('DIRSTR_769'; 'EDM') // Нет данных для формирования отчета.

    Raise(CreateException(''; !Msg; ecInformation))

  EndIf  

  !Запрос="select t.Dok" &

  !Select&"

             from "&!TblName2&" t, MBAnalitSpr avt, MBAnalitSpr pril "&!From&"

             where t.Avtor=avt.Analit

               and t.Pril=pril.Analit "&!Where&"

           order by t.TipD, t.Dok, avt.NameAn, t.DataN"
Шаблон:

SysCode = Application.Connection.SystemInfo.Code

NameGyperServer =Уст("MB_AWebServerName")

HyperDoc = 'http://' &NameGyperServer & '/doc.asp?sys='&SysCode&'&id=' 

RowSize = 13

SymbolLength = 6

Prog.Text = 'Заполнение шаблона XML'

Records = SQL(!Запрос;;;"!")

ColumnCount = SubStringCount(!НазванияКолонок;!ТАБ) 

SheetName =  'Отчет по заявкам эквайринга'

MaxColumnLength = 300

Out =  !НазванияКолонок & cr & cr &  TableColumnType & cr & cr & TableColumnLength

Header = '<Row  ss:Height="'&RowSize&'">' 

foreach str in CSubString(!НазванияКолонок;!ТАБ)

  Header = Header & cr & '  <Cell ss:StyleID="s1"><Data ss:Type="String">'&str&'</Data><NamedCell ss:Name="_FilterDatabase"/></Cell>'

endforeach

Header = Header & cr & "</Row>"

Body = ""

Log = ""

RowCount = 1

Prog.Hide

Max = SubStringCount(Records;"|")

Prog = CreateProgress(;Max) 

Prog.Show 

foreach Record in CSubString(Records;"|")    

  TempColumnLength = ""

  i = 0

  RowNumber = 1

  Cells = ""

  while (i < SubStringCount(Record;"!"))

    i = i + 1

    CurCell = SubString(Record;"!";i)

    CurType = SubString(TableColumnType;"|";i) 

    CurLength = SubString(TableColumnLength;"|";i) 

    if CurType == "Title"

      ID = SubString(CurCell;"(";SubStringCount(CurCell;"("))

      ID = SubString(ID;")";1)    

      Link = HyperDoc&ID

      Cells = Cells & cr &'<Cell ss:StyleID="s2" ss:HRef="'&Link&'"><Data ss:Type="String">'&XMLEscapeSpecialCharacters(CurCell)&'</Data></Cell>'

    endif

    if CurType == "String"

      Cells = Cells & cr &'<Cell ss:StyleID="s3"><Data ss:Type="String">'&XMLEscapeSpecialCharacters(CurCell)&'</Data></Cell>'

    endif

    if CurType == "Number"

      Cells = Cells & cr &'<Cell ss:StyleID="s3"><Data ss:Type="Number">'&CurCell&'</Data></Cell>'

    endif    

    if CurType == "DateTime"

      if Assigned(CurCell)

        XMLDate = FormatDateTimeForXML(CurCell)

        if  SubString(XMLDate;"T";2) == "00:00:00"

          Cells = Cells & cr &'<Cell ss:StyleID="s4"><Data ss:Type="DateTime">'&FormatDateTimeForXML(CurCell)&'</Data></Cell>'

        else

          Cells = Cells & cr &'<Cell ss:StyleID="s5"><Data ss:Type="DateTime">'&FormatDateTimeForXML(CurCell)&'</Data></Cell>'

        endif        

      else

        Cells = Cells & cr &'<Cell ss:StyleID="s3"><Data ss:Type="String"></Data></Cell>'

      endif

    endif  

    CurLength = Макс(CurLength; Length(CurCell) * SymbolLength)

    CurLength = Мин(CurLength; MaxColumnLength)     

    RowNumber = Макс(RowNumber; Truncate((Length(CurCell) * (SymbolLength + 1.25)) / CurLength))

    TempColumnLength = AddSubString(CurLength; TempColumnLength; "|")

  endwhile

  TableColumnLength = TempColumnLength

  Body = Body & cr & '<Row  ss:Height="'&RowNumber*RowSize&'">' & Cells & cr & "</Row>"

  RowCount = RowCount + 1

  Prog.Text = "Обрабатывается " & RowCount & " из " & Max

  Prog.Next

endforeach

Columns = ""

foreach ColumnLength in CSubString(TableColumnLength;"|")

  Columns = Columns &cr& '<Column ss:Width="'&ColumnLength&'"/>'

endforeach

shablon = '

<?xml version="1.0" encoding="windows-1251"?>

<?mso-application progid="Excel.Sheet"?>

<Workbook xmlns="urn:schemas-microsoft-com:office:spreadsheet"

 xmlns:o="urn:schemas-microsoft-com:office:office"

 xmlns:x="urn:schemas-microsoft-com:office:excel"

 xmlns:ss="urn:schemas-microsoft-com:office:spreadsheet"

 xmlns:html="http://www.w3.org/TR/REC-html40">

 <DocumentProperties xmlns="urn:schemas-microsoft-com:office:office">

  <LastAuthor>'&Application.Connection.UserName&'</LastAuthor>

  <Created>'&FormatDateTimeForXML(ServerDateTime())&'</Created>

 </DocumentProperties>

 <ExcelWorkbook xmlns="urn:schemas-microsoft-com:office:excel">

  <ProtectStructure>False</ProtectStructure>

  <ProtectWindows>False</ProtectWindows>

 </ExcelWorkbook>

 <Styles>

  <Style ss:ID="Default" ss:Name="Normal">

   <Alignment ss:Vertical="Bottom"/>

   <Borders/>

   <Font ss:FontName="Arial Cyr" x:CharSet="204"/>

   <Interior/>

   <NumberFormat/>

   <Protection/>

  </Style>

  <Style ss:ID="s1" ss:Name="Шапка">

   <Alignment ss:Horizontal="Center" ss:Vertical="Center" ss:WrapText="1"/>

   <Borders>

    <Border ss:Position="Bottom" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Left" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Right" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Top" ss:LineStyle="Continuous" ss:Weight="1"/>

   </Borders>

   <Font x:CharSet="204" x:Family="Swiss" ss:Color="#FFFFFF" ss:Bold="1"/>

   <Interior ss:Color="#0000FF" ss:Pattern="Solid"/>

  </Style>

  <Style ss:ID="s2" ss:Name="Гиперссылка">

   <Alignment ss:Horizontal="Left" ss:Vertical="Center" ss:WrapText="1"/>

   <Borders>

    <Border ss:Position="Bottom" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Left" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Right" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Top" ss:LineStyle="Continuous" ss:Weight="1"/>

   </Borders>

   <Font ss:FontName="Arial Cyr" x:CharSet="204" ss:Color="#0000FF"

    ss:Underline="Single"/>

  </Style>

  <Style ss:ID="s3" ss:Name="Ячейки">

   <Alignment ss:Horizontal="Left" ss:Vertical="Center" ss:WrapText="1"/>

   <Borders>

    <Border ss:Position="Bottom" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Left" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Right" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Top" ss:LineStyle="Continuous" ss:Weight="1"/>

   </Borders>

   <Font x:CharSet="204" x:Family="Swiss"/>

  </Style>  

  <Style ss:ID="s4" ss:Name="ShortDate">

   <Alignment ss:Horizontal="Center" ss:Vertical="Center" ss:WrapText="1"/>

   <Borders>

    <Border ss:Position="Bottom" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Left" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Right" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Top" ss:LineStyle="Continuous" ss:Weight="1"/>

   </Borders>

   <Font x:CharSet="204" x:Family="Swiss"/>

   <NumberFormat ss:Format="Short Date"/>

  </Style>

  <Style ss:ID="s5" ss:Name="GeneralDate">

   <Alignment ss:Horizontal="Center" ss:Vertical="Center" ss:WrapText="1"/>

   <Borders>

    <Border ss:Position="Bottom" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Left" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Right" ss:LineStyle="Continuous" ss:Weight="1"/>

    <Border ss:Position="Top" ss:LineStyle="Continuous" ss:Weight="1"/>

   </Borders>

   <Font x:CharSet="204" x:Family="Swiss"/>

   <NumberFormat ss:Format="General Date"/>

  </Style>

 </Styles>

 <Worksheet ss:Name="' & SheetName & '">

  <Names>

    <NamedRange ss:Name="_FilterDatabase" ss:RefersTo="=' & "'" & SheetName & "'" & '!R1C1:R1C'&ColumnCount&'" ss:Hidden="1"/>

  </Names>    

  <Table ss:ExpandedColumnCount="'&ColumnCount&'" ss:ExpandedRowCount="'&RowCount&'" x:FullColumns="1" x:FullRows="1" ss:DefaultRowHeight="15">

  '&Columns&cr&Header&cr&Body&cr&'

  </Table> 

  <WorksheetOptions xmlns="urn:schemas-microsoft-com:office:excel">

   <PageSetup>

    <Header x:Margin="0.3"/>

    <Footer x:Margin="0.3"/>

    <PageMargins x:Bottom="0.75" x:Left="0.7" x:Right="0.7" x:Top="0.75"/>

   </PageSetup>

   <Print>

    <ValidPrinterInfo/>

    <PaperSizeIndex>9</PaperSizeIndex>

    <HorizontalResolution>300</HorizontalResolution>

    <VerticalResolution>0</VerticalResolution>

    <NumberofCopies>0</NumberofCopies>

   </Print>

   <Selected/>

   <Panes>

    <Pane>

     <Number>3</Number>

     <ActiveRow>13</ActiveRow>

     <ActiveCol>3</ActiveCol>

    </Pane>

   </Panes>

   <ProtectObjects>False</ProtectObjects>

   <ProtectScenarios>False</ProtectScenarios>

  </WorksheetOptions>

  <AutoFilter x:Range="R1C1:R1C'&ColumnCount&'" xmlns="urn:schemas-microsoft-com:office:excel">

  </AutoFilter>

 </Worksheet>

</Workbook>'

Prog.Hide

#[shablon]

Использование материалы и благодарности:
1. Стандартный отчет Directum: “Сводная таблица по документам”

2. Статья Литвинова Андрея: Использование мастера действий при формировании отчетов 
Данное решение использовалось при разработке проекта: Система управления заявками на получение эквайринговых услуг ЗАО «СНГБ»
